
Resumo

Esse ndo é um livro que ensina programacao, algoritmos ou l6gica. Seu objetivo é servir de manual
para a linguagem G-Portugol e ferramentas relacionadas. Portanto, ele assume que o leitor seja versado
em linguagens de programacao e desenvolvimento de software.



Capitulo 1

Introducao

G-Portugol é um dialeto da linguagem/pseudo-cédigo portugol (ou portugués estruturado), que é muito
usada para descrever algoritmos em portugués, de forma livre e espontanea. Em geral, livros dedicados
ao ensino de algoritmos, l6gica e estruturas de dados utilizam alguma forma dessa linguagem.

A proposta de G-Portugol é disponibilizar uma implementacéo da linguagem portugol, fornecendo fer-
ramentas que oferecam recursos de edicdo, compilagdo, execucédo e depuragdo de programas escritos
nessa linguagem, de forma a favorecer aos estudantes que dao os primeiros passos no aprendizado de de-
senvolvimento de softwares, bem como professores que ensinam disciplinas relacionadas a computagao.
Portanto, seu foco é primariamente didatico.

Encontram-se disponiveis atualmente um compilador, tradutor e interpretador para a linguagem (GPT)
e um ambiente visual simples (GPTEditor) que permite a edigcdo, execucdo e depuracdo de programas
escritos em G-Portugol.

A seguir sao apresentados os assuntos abordados nos capitulos seguintes:

e Capitulo 2: pretende discutir as caracteristicas gerais da linguagem.

e Capitulo 3: aborda a programacdo em G-Portugol, suas estruturas e recursos, utilizando exemp-
los ilustrativos e comparando com linguagens populares. Embora o capitulo trate da programacao,
ele ndo tem como objetivo explicar programacdo em si ou a teoria/histéria por tras das estruturas
abordadas.

e Capitulo 4: trata da implementagao da linguagem. Discute as decisdes de design e recursos ofereci-
dos por G-Portugol.

e Capitulo 5: descreve o programa GPT.

No apéndice deste livro pode ser encontrado a gramatica da linguagem G-Portugol.

Capitulo 2

Caracteristicas Gerais

A linguagem em si ndo difere fundamentalmente ou apresenta novidades em relagdo ao uso popular de
portugol. Semelhante a linguagens como Pascal e C, é uma linguagem imperativa, com comandos de
controle de fluxo, manipulacado de dados e recursos de entrada e saida basicos. A linguagem em case
sensitive, 0 que significa que, por exemplo, uma fungao chamada “leia” é diferente de uma funcdo chamada
“Leia”.

2.1 Tipos de dados

Tipos primitivos como inteiro e literal sdo suportados. Literais séo tipos integrais, e néo, tipos construidos
pelo usuario. A tabela 2.1 apresenta os tipos com respectivos exemplos de dados:

Tipos primitivos
Tipo Exemplos
inteiro 19
real 2.5
caractere | 'a’
literal “‘uma frase”
l6gico verdadeiro

Tabela 2.1: Tipos de dados primitivos

Com esses tipos, pode-se criar conjuntos como vetores ou matrizes “n” dimensionais. Tipos mais
complexos ndo sao suportados.

2.2 Estruturas de controle

O conjunto de estruturas de controle sdo os mais primitivos. Uma estrutura condicional (se/senao) e duas
estruturas de repeticdo (enquanto/para) sdo suportadas. Embora a estrutura “para” seja uma especializa-
¢ao de um lago simples como o “enquanto”, ela foi implementada, visto que uma variedade de livros e
muitos professores os discutem.

2.3 Subprogramas (funcoes)

Subprogramas sao como fungdes em C. Podem receber qualquer nimero de parametros, sejam tipos prim-
itivos ou vetores/matrizes, e podem retornar apenas valores primitivos. Entretanto, ndo é permitido declarar

Capitulo 2. Caracteristicas Gerais

Estruturas de repeticao
Estrutura Tipo
se/entdo/sendo/ | condicional
enquanto repeticao
repita repeticao
para repeticao

Tabela 2.2: Estruturas de repeticao

funcdes aninhadas ou fungdes com parametros variaveis. Vale ressaltar que passagem de parametros é
sempre feita por valor.

Capitulo 3

Programando em G-Portugol

A proposta desse capitulo € mostrar os elementos da linguagem G-Portugol usando programas como
exemplos. Nao é um capitulo sobre como programar, nem tem como intuito ensinar algoritmos. Portanto,
€ esperado que o leitor seja familiarizado com programacao.

3.1 Ola Mundo

Mantendo a tradigdo, vamos criar nosso primeiro programa. O proposito dele sera exibir na tela o texto
"Ola mundo". Crie um arquivo chamado "olamundo.gpt"com o seguinte conteudo:

Programa 1 “Ola Mundo” em G-Portugol.
/%

Nosso primeiro programa

*/

algoritmo olamundo;

inicio
imprima("0l4 mundo!");
fim

Apods salvar o arquivo, digite o seguinte na linha de comando:

No Linux:
$ gpt -o olamundo olamundo.gpt
No Windows:
\> gpt -o olamundo.exe olamundo.gpt

Esse comando compila o algoritmo e salva o arquivo binario resultante como “olamundo” (ou “ola-
mundo.exe”) no diretorio atual. Se a opgdo “-0 <arquivo>" nao for informada, o GPT criara o executavel
usando o nome do algoritmo. Ap6s executar o programa criado, é exibido o texto “Ola mundo!” na tela.

Como j& deve ter assumido, comentérios ao estilo C (/**/) e C++ (/) sdo permitidos e o0 que estiver entre
aspas duplas (") é tratado como uma string ou constante literal. Escapes s@o permitidos como EOL (\n),
tabulacéo (\t) dentre outros. N&o é permitido a concatenagdo em multiplas linhas como na linguagem C:

Capitulo 3. Programando em G-Portugol

imprima("Isso & uma "
"Concatenag&o"); //erro!

.,y £

Outro detalhe é a pontuagdo. Da mesma forma que em C, o “;” é utilizado como finalizador de enuncia-
dos. Se for necessario utilizar uma aspa dupla em um literal, deve-se utilizar o escape:

imprima("Isso & literal com aspas duplas ($\backslash$") ");

3.2 Variaveis

A declaragao de variaveis deve ser feita dentro de um bloco especifico, que deve aparecer logo apdés a
declaragéo do algoritmo. O bloco é iniciado pela palavra-chave “variaveis” (sim, com acento) e termina
com a palavra chave “fim-varidveis”. Pelo menos uma varidvel deve ser declarada dentro do bloco (embora
0 bloco em si seja opcional) e apenas um bloco em escopo global deve ser declarado. Eis um exemplo
para estudo:

Programa 2 Declaragao de varigveis globais.
algoritmo teste_variaveis;

variaveis
X : inteiro;
nome : literal;
fim-variaveis

inicio
fim

3.2.1 \Variaveis primitivas
Variaveis primitivas sdo declaradas seguindo o seguinte modelo:

<identificador> [, identificador]* : <tipo>;

Isso €, um ou mais identificadores separados por virgula, seguido de um “”, seguido de um tipo,
seguido, finalmente, de “;”. Como pode-se notar, é parecido com Pascal. Assim que um programa é
executado, todas as variaveis, primitivas ou vetores/matrizes, declaradas s&o iniciadas com um valor nulo
ou “0” automaticamente. Vale ressaltar que constantes (const, final, etc) ndo sdo suportados. Os tipos
primitivos suportados encontram-se na tabela 2.1

A fim de explorar melhor os aspectos da declaragdo, seguem-se alguns comentarios a respeito do
ultimo exemplo (programa 2)

e Observe que o nome do algoritmo (“teste_variaveis”) ndo tem acento. Se declarar o algoritmo como
“teste_variaveis” e tentar compilar o cédigo, o seguinte erro sera exibido:

Linha: 1 - “teste_variaveis” nao pode ter caracteres especiais.

Portanto, identificadores (nomes de varidveis, fungbes e do algoritmo) ndo podem ter acentos ou
caracteres especiais como $, #, etc. A definicdo de um identificador em G-Portugol é equivalente ao
das linguagens populares: uma letra (a-z ou A-Z) seguido de qualquer niUmero de letras ou nimeros.
Finalmente, underlines (_) sdo permitidos. Cedilhas, portanto, também nao formam identificadores
vélidos.

¢ O segundo ponto, € a palavra-chave “variaveis”: ela tem acento, e isso € permitido e obrigatorio.

Capitulo 3. Programando em G-Portugol

e O terceiro, é a definicdo do bloco e sua (falta de) semelhanca com o Pascal. Todas os blocos em
G-Portugol tentam seguir o formato “nome/fim-nome”, em favor da uniformidade e em detrimento de
excecgoes linguisticas que confundem os estudantes.

¢ E, finalmente, o quarto ponto é a auséncia de cédigo entre “inicio” e “fim”. O programa nao é obrigado
a ter enunciados ou comandos.

Para maiores detalhes, veja o capitulo 4, sobre a implementagao da linguagem.

3.2.2 Vetores e matrizes (conjuntos)

Vetores e matrizes “n” dimensionais de tipos primitivos sdo suportados. Um exemplo de declaragcédo de
uma matriz:

variaveis
bitset : matriz[10] de 1légicos;
quadr : matriz[4][4] de inteiros;
fim-variaveis

O tipo do vetor ou matriz é dado pelo nome do tipo no plural (“inteiros” para tipo inteiro, “literais” para
tipo literal, etc). Os subscritos na declaragéo (delimitados por “[]”) indicam o tamanho da matriz, e sua
dimensao é informada pelo nimero de subscritos. Portanto, “bitset” é um vetor de 10 valores légicos,
enquanto “quadr” é uma matriz bidimensional, onde as duas dimensées tem tamanho 4.

E importante observar que matrizes sdo “0 based”, isso &, iniciam no indice 0 e seus indices sdo sempre
inteiros positivos. Logo, a matriz “bitset” pode ser usada do indice 0 até o indice 9 (inclusive). Seu indice
10 ndo é valido e seu uso podera acarretar em erros de execugao (runtime errors). Matrizes, assim como
varidveis de tipos primitivos, sdo inicializadas com o valor “0” ou “nulo” em todas as suas posigdes. Quando
usadas como argumentos de fun¢des, matrizes sédo passadas por valor.

“As pessoas sdo dividas em dois grupos: aquelas que comegam a contar a partir do '0’, e
aquelas que ndo.”

— Anbnimo

3.3 Estruturas condicionais

Por enquanto, apenas a estrutura se/entdo/sendo é suportada. Essa e as demais estruturas utilizam
expressoes, que sao avaliadas para que uma decisdo seja tomada (repetir execugao, selecionar bloco
de instrugdes, etc). Qualquer expressao pode ser avaliada como expressao légica. Expressées numéricas
de valor “0” sdo avaliadas como falso. Demais valores numéricos sédo avaliados como verdadeiro. Valores
literais nulos, da mesma forma, sdo avaliados como falso, e demais textos, como verdadeiro. Para maiores
detalhes sobre expressodes, veja a se¢ao 4.3.3.

enquanto x faga //depende do valor de x
enquanto '"nome" = "nome" faga //avalia como verdadeiro
enquanto verdadeiro faga //avalia como verdadeiro
enquanto 3+5 faca //avalia 8 como verdadeiro
enquanto '"nome" faga //avalia como verdadeiro
enquanto 0 faga //avalia 0 como falso
enquanto "" facga //avalia como falso
enquanto ’’ faga //avalia como falso

O programa 3 ilustra um algoritmo simples que verifica se o usudario € maior de idade.

Alguns pontos a serem considerados:

Capitulo 3. Programando em G-Portugol

Programa 3 Exemplo de programa que utiliza estrutura condicional.

algoritmo idade;

variaveis

idade : inteiro;
nome : literal;

fim-variaveis
inicio
imprima("Digite seu nome:");
nome := leia();
imprima(nome, ", digite sua idade:");

idade := leia();

se idade >= 18 entdo

se idade < 60 entdo
imprima("adulto!");
sendo
imprima("ancido", ’!’);
fim-se

sendo

imprima("menor", "!");

fim-se

fim

O nome do algoritmo é “idade”, assim como o nome de uma variavel. Nao ha conflitos.

i

O operador de atribuicao é o pascal-like “:=".
A funcgéo interna “leia” pode ser usada por variaveis primitivas de qualquer tipo.

A fungéo “imprima” recebe um numero variavel de argumentos de qualquer tipo primitivo, sendo que
pelo menos um argumento deve ser passado, que podem ser constantes literais (nimeros, textos
entre aspas, caracteres entre aspas simples,...), variaveis primitivas ou indices de vetores/matrizes.
Os valores séo concatenados e adicionados de um caractere EOL (end of line, ou fim de linha).

J& vimos que strings/literais constantes sdo denotados por texto entre aspas duplas ("). Tal qual em
C, um caractere entre aspas simples (') € um caractere constante que, também, permite escapes
para representar caracteres como EOL (\n’).

(LT

Assim como a palavra-chave “varidveis”, “entdo” e “sendo” devem ser acentuadas.
Paréntesis ao redor da expressao da estrutura “se/entdo” sao opcionais.

Como em “variaveis/fim-variaveis”, blocos “se” tem seus respectivos e obrigatorios “fim-se”. Nao ha a
opcao de ignorar a definigdo do bloco quando apenas um enunciado é usado, como na linguagem C

onde o uso de “{}” é opcional. Também, ndo ha imposicdes sobre a indentacdo. O programa 4 ilustra
esse assunto.

Capitulo 3. Programando em G-Portugol

Programa 4 Uso incorreto de estrutura condicional.
//cddigo invalido: faltando fim-se
algoritmo se_invalido;

inicio
se x = 2 entéo
imprima("ok");
imprima("estou dentro ou fora do bloco \"se\"7");
fim

3.4 Estruturas de repeticao

3.4.1 A estrutura “enquanto”

A estrutura “enquanto” é a mais basica e comum das estruturas de repeticdo. Seu funcionamento é o
mesmo que em outras linguagens populares, onde um conjunto de instrugdes é executado repetidamente
enquanto o valor I6gico de uma dada expresséao for avaliado como “verdadeiro”.

Programa 5 Exemplo de programa que utiliza a estrutura “enquanto”.
algoritmo fatorial;

variaveis
res : inteiro;
fat : inteiro;
X : inteiro;
fim-variaveis

inicio
imprima("Digite um nimero:");
x := leia();

fat := x;
res := 1;

enquanto x <> 0 faga
res := res * Xx;
X = x - 1;
fim-enquanto

imprima("fatorial de ",fat," & igual a ",res);
fim

Assim como na estrutura “se/entao”, parénteses em volta da expressado sao opcionais e as expressdes
seguem as mesmas regras.

3.4.2 A estrutura “repita”

A estrutura “repita” tem o funcionamento semelhante a “enquanto”. A diferenca esta na ordem da avaliacao
da condicdo de repeticdo: “enquanto” avalia a condi¢cdo antes de iniciar o lago, e “repita” avalia ao final do
bloco.

Capitulo 3. Programando em G-Portugol

Programa 6 Exemplo de programa que utiliza a estrutura “repita.”
algoritmo fatorial2;

variaveis
res : inteiro;
fat : inteiro;
X : inteiro;
fim-variaveis

inicio
imprima("Digite um nimero:");
fat := leia();

res := 1;

x := fat;

repita
res := res * X;
X =x - 1;

até x <= 0;

imprima("fatorial de ",fat," é& igual a ",res);
fim

3.4.3 A estrutura “para”

A estrutura “para” € uma especializacdo da estrutura “enquanto”, e costuma ser ensinada em cursos de
programagédo. Sua sintaxe é semelhante ao que se vé em literatura sobre algoritmos e estruturas de dados,
entretanto, € uma sintaxe diferente de linguagens populares como C e Java.

A estrutura “para” tem a seguinte forma:

para <variivel> de <express&o> até <expressfo> [passo <inteiro>] faga
[lista de comandos]
fim-para

Onde:

e “variavel” deve ser uma variavel numérica;

e “‘expressao” deve ser uma expressao que tem seu valor avaliado como numérico;
e “passo”, se existir, deve ser seguido por um inteiro constante.

As expressdes de/até controlam os valores que a variavel numérica tera no inicio e no fim do lago,
respectivamente. Tanto o controle da frequéncia, quanto a decisdo de incrementar ou decrementar a
variavel de controle é feita pelo termo opcional “passo”, e seu valor padrao é 1. Por exemplo, para iterar o
valor de uma varidvel numérica “x” de 0 até 10, escreve-se:

para x de 0 até 10 faga
//comandos. ..
fim-para

Da mesma forma, para uma iteragdo decrescente, de 2 em 2, escreve-se:

Capitulo 3. Programando em G-Portugol

para x de 10 até O passo -2 faga
//comandos. . .
fim-para

Programa 7 Exemplo de programa que utiliza a estrutura “para”.
algoritmo fatorial;

variaveis
res : inteiro;
fat : inteiro;
X : inteiro;
fim-variaveis

inicio
imprima("Digite um nimero:");
fat := leia();

res := 1;

para x de fat até 1 passo -1 faga
res := res * X;

fim-para

imprima("fatorial de ",fat," & igual a ",res);

fim

Em comparagéo com a estrutura “for” de linguagens com sintaxe baseadas em C, ha diferencas nao s6
de sintaxe, mas de implementacdo. Um “for” que itera sobre uma variavel numérica de 0 até (incluindo) 10,
ao sair do lago, o valor dessa variavel sera 11. Em G-Portugol, a variavel tera o valor 10 ao sair do lago.
Essa diferenga acontece porque a sintaxe do “para” induz a esse comportamento, diferente da sintaxe do
“for”, que tem um aspecto de mais baixo nivel.

//cddigo em C
for(x = 0; x <= 10; x++);
printf ("%d", x); //imprime "11"

//cddigo equivalente em G-Portugol
para x de 0 até 10 faga
fim-para

imprima(x); //imprime "10"

Ademais, da mesma forma que o “for”, é possivel que a variavel de controle tenha seu valor alterado
pelos comandos aninhados. Isso permite que o lago seja encerrado prematuramente, como também é
comum em estruturas como “enquanto”. A utilidade dessa técnica esta no fato de G-Portugol nao incorpo-
rar mecanismos para refinar o controle de lagos (como “break” e “continue”, encontrados em linguagens
populares).

Capitulo 3. Programando em G-Portugol

3.5 Funcoes

Subprogramas em G-Portugol sdo implementados no modelo de fungdes, que podem ter zero ou mais
parametros de qualquer tipo, tanto primitivos quanto complexos (vetores e matrizes). Opcionalmente, elas
podem definir valores de retorno, que deve ser de tipo primitivo. Tanto o retorno de dados como a passagem
de argumentos sao feitos por valor.

Para retorno prematuro ou retorno de dados, a palavra chave “retorne” é usada. Para fungdes que
retornam dados, “retorne” deve ser seguido de um operando, que é uma expressao cujo valor deve ser
compativel com o tipo da funcéo. Ja fungdes que nao declaram um tipo de retorno expliticamente (equiva-
lente a fungdes de retorno “void” em C), “retorne” deve ser usado sem operando.

Programa 8 Exemplo de algoritmo que utiliza fungdes.
algoritmo fatorial_recursivo;

variaveis
X : inteiro;
fim-variaveis

inicio

imprima("Digite um namero:");

x := leia();

imprima("fatorial de ",x," & igual a ",fatorial(x));
fim

fungdo fatorial(z:inteiro) : inteiro
inicio
se z = 1 entéo
retorne 1;
sendo
retorne z * fatorial(z-1);
fim-se
fim

3.5.1 Funcoes internas

Como ja foi visto em exemplos anteriores, G-Portugol oferece duas fungdes internas: “leia” e “imprima”,
que permitem uso basico de entrada e saida, respectivamente.

A funcao “leia” nao recebe argumentos e retorna o valor lido da entrada padrao (“STDIN”), o que sig-
nifica, geralmente, ler os dados que o usuario digitar no teclado, seguido do caractere “nova linha” (em
geral, associado a tecla “Enter” no teclado). O tipo de dado retornado por “leia” é implicitamente convertido
para o tipo primitivo exigido no contexto em que ela € usada.

A fungao “imprima” recebe um ndmero variavel de argumentos de qualquer tipo primitivo, sendo que
pelo menos um argumento deve ser passado. Os valores passados como argumentos sdo convertidos
para texto, concatenados na ordem definida e enviados para “STDOUT” (em geral, associado ao display
ou monitor). N&o ha retorno de valor para esta fungao.

Capitulo 3. Programando em G-Portugol

Programa 9 Exemplo de usos das fun¢des internas “leia” e “imprima”.
algoritmo io;

variaveis
c: caractere;
i: inteiro;
r: real;
1: literal;
z: loégico;

mat: matriz[2] de inteiros;
fim-variaveis

inicio
imprima("digite um caractere");
c := leia();
imprima("Digite um nimero inteiro");
i := leia();
imprima("Digite um ndmero real:");
r := leia();
imprima("Digite um texto:");
1 := leia();
imprima("Digite um valor 16gico (\"verdadeiro\" ou \"falso\"), um ndmero ou um texto:");
z := leia();

" " ,i, ", real: "

imprima("caractere: ",c,", inteiro: " ,r,", texto: ",1, ", logico: ", z,"\n");

fim

Capitulo 4

Implementacao da linguagem
G-Portugol

4.1 Introducao

Ao definir uma linguagem de programacao voltada para o ensino de ldgica e algoritmos, varios aspectos
devem ser considerados. Ao contrario de linguagens de produgéo, ndo ha preocupagcdes como o poder
expressivo da linguagem, dicionario em inglés, acesso a recursos de sistema, etc. A preocupacgao central
estd em oferecer uma ferramenta que:

o reflita os processos computacionais, exigindo o minimo de conhecimento e experiéncia do estudante;
e evidencie os processos relacionados com o desenvolvimento de softwares;

e estimule a abstracao e raciocinio légico do estudante.

4.2 A linguagem

O ponto fundamental que guia as diretrizes da linguagem G-Portugol é seu propésito educacional: ela deve
expressar processos computacionais de forma que um leigo os compreenda sem enfatizar a si mesma.
Isso é, a linguagem em si deve chamar o minimo de atengao possivel, fazendo com que a compreensao
dos processos computacionais seja tao natural quanto ler sua descrigédo informal, ou ndo-estruturada.

Esse objetivo encontra restricées, quando se leva em consideragao a natureza das linguagens artificiais
e o uso e forma popular atual da linguagem portugol, principalmente em literaturas. E de interesse que
G-Portugol ofereca compatibilidade com essas formas, o que pode gerar conflitos quanto a decisdes de
design e restringir suas caracteristicas. Portanto, embora as diretrizes marquem a base do design, muitas
vezes, elas devem ser sacrificadas ou ajustadas para incorporar formas populares.

E interessante ressaltar que criar uma linguagem totalmente nova, que utiliza outros paradigmas como,
por exemplo, orientacdo a objeto, possa ser interessantes e, talvez, mais eficientes como ferramentas de
ensino, mas G-Portugol esta, no momento, comprometida com a compatibilidade.

A seguir, alguns topicos serao comentados quanto as diretrizes por tras das formas Iéxicas e gramati-
cais da linguagem G-Portugol.

4.2.1 Diretrizes para o design da linguagem
A linguagem deve ser totalmente em portugués e deve respeitar acentuacoes

Linguagens de programacao, em geral, ndo se utilizam de caracteres especiais (ex. caracteres acentuados,
cedilhas, e outros que ndo pertencem ao alfabeto inglés) para definicdo de seu dicionario, visto que sao
baseadas na lingua inglesa. Portanto, letras acentuadas ndo séo consideradas.

A decisao de incorporar palavras que respeitam a lingua portuguesa é importante, visto que modificar
a linguagem de forma a se afastar de sua lingua natural (o portugués) pode evidenciar excessdes as quais
forcariam os usuarios (estudantes e professores) a se ater mais com o estudo da linguagem do que com o
estudo da disciplina em questao. Isso é, a auséncia de acentos, por exemplo, obriga o usuario a aprender
seus termos excepcionais. Além do mais, 0 uso de termos como “nao”, chama a atengao constante do
usuario para o fato de a palavra nao estar acentuada, o que costuma desviar atencdo do estudo.

Consequentemente, o uso de acentos permite que a linguagem seja o mais préximo do portugués
quanto for possivel, apoiando a regra de nao chamar atencgao para si. Ademais, o cédigo fica mais legivel
e permite uma leitura mais agradavel.

Além dos acentos, é exigido que as palavras-chave usadas sejam completas ou por extenso, sem
permitir abreviacdes (ex. “proc”, “func”, “char”, “int”, ...), o que dificulta a leitura de programas por um leigo.

Ha também decisdes quanto a forma verbal de comandos e fungdes. Em geral, na literatura, os verbos
nos algoritmos sdo expressos no imperativo, visto que a linguagem é caracterizada como imperativa. Por
outro lado, vale notar que, mesmo em linguagens imperativas, € comum ver programas que utilizam termos
em portugués usando verbos no infinitivo.

Mesmo com essa perspectiva em vista, G-Portugol se utiliza de verbos no imperativo para seus termos,
de forma a se aproximar das formas utilizadas nas literaturas sobre algoritmos.

“Programas devem ser escritos para pessoas lerem e, apenas incidentalmente, para maquinas
executarem.”

— Abelson e Sussman

A linguagem deve ser simples, uniforme e coerente

Essa diretriz diz respeito a evitar excessdes entre as formas gramaticais e comportamentos semanticos na
medida do possivel. Isso significa ndo sé que as estruturas e comandos devem ser simples, claros e ébvios,
como também, sua forma e comportamento devem ter um principio em comum, evidenciando um formato
uniforme. Naturalmente, a linguagem portugol ja detém uma parcela relevante de caracteristicas que se
acomodam nessas diretrizes, no entanto, alguns detalhes ainda merecem atencéo, como a representacao
de blocos estruturais.

A simplicidade implica em evitar construgdes e recursos que desviam a atengao do propdsito original
para detalhes da linguagem. Isso €, apenas recursos essenciais devem fazer parte de seu nudcleo. Por-
tanto, em um primeiro momento, ndo é interessante equipar a linguagem com recursos como alocagao
dinamica de meméoria, controle de threads, etc.

Deve ser configuravel naquilo que diz respeito a diferentes abordagens de ensino e estilo do pro-
fessor

Essa diretriz diz respeito a permitir que a linguagem mude em certos aspectos (ou disponibilizar meios
simples para realizar essas mudancgas) de acordo com o gosto do professor, ou como modo de adaptar
a forma de portugol de uma dada literatura. Dialetos e formas da linguagem portugol variam de literatura
para literatura, e pode ser interessante permitir que a linguagem se adapte ou oferecer diferentes formas
da linguagem.

Entre os aspectos que podem ser adaptados com facilidade se encontram os termos Iéxicos (palavras-
chave, operadores,etc), e algumas formas de expressar estruturas, como algumas declara¢des e enuncia-
dos.

Por exemplo, pode ser interessante que, para se declarar uma variavel, ndo se use um bloco iniciado
por “variaveis” e terminado por “fim-variaveis”, mas como um bloco parecido com a declaragdo de variaveis
em Pascal, ou até em C (que sequer exige delimitacdo de bloco de variaveis).

Isso pode ser interessante quando se quer, por exemplo, ensinar uma linguagem especifica como
Pascal, fazendo com que G-Portugol possa usar estruturas e operadores semelhantes a esta linguagem,
de forma a oferecer uma transicao mais direta e que aproveita melhor os conhecimentos do estudante.

Atualmente, o programa GPT nao implementa mecanismos para esse tipo de mudancga, sendo necessario
modificar seu codigo fonte e recompila-lo.

4.3 Formato Estrutural

Um programa escrito em G-Portugol tem o seguinte formato:

e declaracao do algoritmo
e declaragdo de variaveis globais
e bloco principal

e declaracao de funcdes

A seguir, alguns pontos serao discutidos a respeito do formato da linguagem.

4.3.1 Declaracao do algoritmo

A declaracado do algoritmo nao influencia o programa ou sua execugao, visto que &€ apenas uma sessao
informativa no cédigo que, embora ndo seja um comentério, tem o mesmo efeito pratico. A ado¢do dessa
declaracao pode ser discutida, e foi escolhida por ser bastante utilizada em literaturas.

4.3.2 Declaracao de variaveis globais

Como visto no capitulo 2, essa declaracdo é opcional, e seu formato pode ser visto no exemplo a seguir:

variaveis
X : inteiro;
fim-variaveis

Esse formato difere de linguagens como Pascal e C. Em Pascal, o bloco ndo tem um delimitador final
(como “fim-variaveis”) e em C, n&o existe qualquer delimitador.

Delimitar o bloco permite maior consisténcia com outras formas gramaticais como os blocos de co-
mando e estruturas de controle, e torna o cédigo mais claro e explicito, embora adicione construgcdes
redundantes.

Os tipos primitivos englobam os tipos mais basicos das linguagens populares. No momento, agregados
heterogéneos como, por exemplo, estruturas/registros (“struct” em C) ndo sdo suportados. Para uma lista
dos tipos suportados, veja a tabela 2.1.

4.3.3 Bloco Principal

As linguagens de programacéao devem, de alguma forma, oferecer um “entry point” (ponto de entrada), de
onde se inicia a execucdo do programa. O ponto de entrada pode ser uma fungdo ou um bloco anénimo.
Na literatura, em geral, o bloco principal € delimitado pelos termos “inicio’e “fim” e G-Portugol segue essa
convengdo. Essa decisdo mantém um nivel satisfatério de coeréncia com o bloco de variaveis globais e
estruturas de controle, embora ndo sejam intimamente relacionados.

Vale ressaltar que nao se faz imposicdo sobre a identagéo do codigo. Esse recurso pode ser vantajoso
no ensino, promovendo a clareza de c6digo, portanto, sua implementacao pode ser discutida para versdes
futuras.

Embora a sintaxe seja bem diferente, o bloco principal pode ser visto como uma fungao (como main na
linguagem C). Ha, portanto, a possibilidade de retornar um valor inteiro através do comando “retorne”.

Programa 10 Bloco principal com comando retorne.
algoritmo bloco_principal;

inicio
imprima("Retornando valor do bloco principal");
retorne 42;
imprima("Essa linha n#o serd impressa");

fim

Estruturas de Controle

Estruturas de controle sdo formadas por um cabecgalho seguido por um bloco de comandos e terminados
por um delimitador final. Embora os delimitadores do bloco ndo sejam “inicio” e “fim”, ha um grau de
semelhanga mantido: o delimitador inicial € omitido (uma vez que o cabegalho é entendido como delimita-
dor inicial) e o delimitador final € o termo “fim-" seguido do nome da estrutura.

No cabecalho das estruturas como “enquanto” e “se”, as expressdes nao precisam ser delimitadas por
parénteses.

Expressoes

Expressdes sdo operagdes que sintetizam, ao final, um valor. Em geral constituem valores ou operagdes
aritméticas com um ou mais termos que podem ser variaveis, constantes ou chamadas a fungbes e sao
usadas em atribuicoes, estruturas de controle e subscritos de matrizes e vetores. Enunciados (como
atribuicées) ndo podem ser avaliados como expressoes.

Constantes sao valores inline, e cada tipo de dado tem uma forma de ser representada.

e Inteiros: Podem ser representados em base decimal, hexadecimal, octal e binaria. Representagdes
decimais sao formadas, opcionalmente, por um sinal (“+” ou “-”) seguido de um mais algarismos(ex.“120”,
“+57, e “-2”). Representacdes hexadecimais sdo representadas com o prefixo “0x” ou “ 0X”, seguido
de algarismos entre “0” e “9” e letras entre “a” e “f” ou “A” e “F” (ex. “OxF1A5”). Representagdes oc-
tais sao representadas com o prefixo “0c” ou “0C”, seguido de algarismos entre “0” e “7” (ex. “0c61”).
Finalmente, representacdes binarias sdo formadas pelo prefixo “Ob” ou “0B”, seguido de algarismos

“0” e “1”7 (ex. “0b101”).

“ . n

¢ Reais: sdo representados por, opcionalmente, um sinal (“+” ou “-”), seguido de algarismos separados
por um “” como “-1.2345”.

e Caracteres: sao representados por um Unico simbolo entre aspas simples. Alguns caracteres espe-
ciais sdo representados com escape (“\”) seguido de simbolo identificador. Esses caracteres séo o
LF (“\n”), CR (“\r") e barra invertida (“\\”). A auséncia de simbolos entre as aspas simples indica um
caractere “nulo”. Internamente, caracteres sao representados como nimeros inteiros, o que permite
sua compatibilidade numérica.

e Literais: sdo representados por um conjunto de caracteres entre aspas duplas. Eles podem conter
caracteres especiais como “\n” e devem ser definidos em apenas uma linha de cédigo. Valores literais
sd0 0s Unicos que ndo tem uma representacdo numérica, impedindo sua participagdo em expressoes
com operadores aritméticos (soma, divisdo, etc). Comparagdes de igualdade para valores literais
sdo feitas caractere por caractere em case sensitive. Portanto, a expressao “portugol” = “Portugol”
é avaliada como falsa. J& comparacbes de grandeza sao feitos calculando o nimero de caracteres
que compdem os valores literais. Entao, a expressao “maria” > “josé” é avaliada como verdadeira.

e Logicos: sao representados pelas palavras “verdadeiro” e “falso”. Numericamente, qualquer valor
diferente de “0” representa o valor verdadeiro e “0” representa o valor falso.

A precedéncia de operadores é mostrada na tabela 4.1 (da menor precedéncia para a maior) e pode
ser explicicamente modificada com o uso de parénteses.

Precedéncia de Operadores
Operador Nome
ou OR logico
e AND légico
| OR binario
" XOR binario
& AND binario
=, <> operadores igual e diferente
>, >=, <, <= | operadores relacionais maior, maior ou igual, menor, menor ou igual
+, - operadores aritméticos soma e subtragao
1, % operadores aritméticos multiplicagao, divisdo, médulo
+, -, , nao, () | operadores unarios positivo, negativo, NOT binério, NOT Iogico, parénteses

Tabela 4.1: Precedéncia de Operadores

Em G-Portugol, ndo ha coercéo ou casting de tipos expliticamente. Todos os tipos numéricos (inteiro,
real, 16gico e caractere) sdo compativeis entre si. E importante ressaltar que expressdes envolvendo
igualdade ou diferenca de valores reais ndo sdo apropriadas dado a forma como esses valores sédo repre-
sentados internamente. Tais comparagées podem ter resultados imprevisiveis, e até diferentes, ao serem
executados nos modos compilado, interpretado ou traduzido.

Existem dois casos em que ocorre uma coer¢ao implicita. O primeiro caso ocorre durante a avaliagao
de uma expressao que tem operandos de tipos diferentes (mas compativeis entre si), onde nao ha perda de
dados e um dos termos é promovido para o tipo do seu termo complementar. Por exemplo, na expressao
“2 +1.5”, 0 termo “2” é promovido para o tipo “real”, tendo o valor “2.0” antes que a soma seja processada.
Da mesma forma, sendo “x” uma variavel de tipo real, a expressao “x := 5/ 2” atribuira a “x” o valor “2.00".
Esse comportamento ndo muito 6bvio € similar ao da linguagem C, onde “5 / 2” é avaliado como uma
divisdo de inteiros, onde apenas o valor inteiro final € relevante. Para obter o resultado real, a expressao
deve ser “5.0 / 2”, para informar ao compilador que a divisdo usara valores reais, produzindo a promogao
de tipos mencionada anteriormente.

O segundo caso em que ocorre uma coergao implicita € discutido na segao 4.3.4.

4.3.4 Atribuicoes

Atribuicées permitem a manipulacdo de valores na memoéria. Em G-Portugol, como ja foi visto, € usado
0 operador “:=", onde o termo a esquerda ou “/value” deve ser uma variavel primitiva ou indice de uma
matriz/vetor, e o termo a direita, uma expressdo que, quando avaliada, tem seu tipo compativel com o
Ivalue.

Pode haver coercao de tipos durante a atribuicao, quando o resultado da expressao é de um tipo difer-
ente (mas compativel) do tipo de Ivalue. E possivel que dados sejam comprometidos, por exemplo, tendo
uma expressao avaliada como real sendo atribuida a uma variavel de tipo inteiro (o valor serd truncado).

Matrizes ndo sdo aceitas como /value, como pode ser visto no programa 11.

4.4 Funcoes

Fungbes sdo os subprogramas de G-Portugol. Sao definidas ap6s o bloco principal e podem receber
argumentos e retornar valores. Em tempo de execucgéo, as fungbes criam um novo escopo sobreposto
ao escopo principal. Isso permite recursos como recursdo e possibilita que variaveis locais (no escopo
da fungdo) tenham o mesmo nome que variaveis globais, onde essas Ultimas ficam “escondidas”, isso €,

Programa 11 Uso incorreto de matrizes.
algoritmo atribuicao_de_matrizes;

variaveis
ml : matriz[2] de inteiros;
m2 : matriz[2] de inteiros;
fim-variaveis

inicio
ml := m2; //erro
imprima(ml); //erro
fim

incapazes de serem acessadas enquanto o escopo durar. Ressalta-se que ndo ha suporte para fungdes
aninhadas, isso é, fun¢des declaradas dentro de fungdes.

O retorno de dados é feito por meio da instrugéo “retorne” e o valor de retorno (se houver) deve ser
compativel com o tipo da funcéo. Esse tipo ndo pode ser um tipo agregado como matrizes e vetores.

Tanto a passagem de argumentos quanto o retorno é feito por valor, isso é, a cépia do valor é feita, ao
invés de a copia de enderego ou passagem por referéncia.

As variaveis locais de uma fungéo sao formadas por seus parametros e pelas variaveis declaradas em
seu escopo. A declaragao de variaveis locais é feita entre o cabegalho da funcao e a palavra-chave “inicio”,
portanto, ndo é uniforme em relacdo a declaracao global, onde se usa as palavras-chave “variaveis” e
“fim-variaveis”.

A declaracédo dos parametros da fungado também néo segue estritamente o formato de declaracdo de
variaveis. O programa 12 ilustra a declaragao de uma fungao.

Programa 12 Exemplo de usos de fungdes.
algoritmo exemplo_funcao;

inicio
imprima (soma(2,2)) ;
fim

func8o soma(x: inteiro, y: inteiro) : inteiro
res : inteiro;

inicio
res := X + y;
retorne res;

fim

4.5 Funcoes internas

G-Portugol oferece duas fungdes internas para manipulagao basica de entrada e saida. Ambas as fungoes
tem comportamentos excepcionais quando comparadas com as fungdes de usuario. Essas fungdes sao
discutidas na sessao a seguir.

4.5.1 A funcao “imprima”

A funcdo imprima tem como objetivo imprimir texto no dispositivo de saida. Ela recebe um namero variavel
de argumentos (pelo menos um), onde cada argumento deve ser um valor primitivo, e os imprime em
sequéncia. Ao final, é impresso um caractere de nova linha (LF) e a fungao retorna.

Essa funcdo se comporta de forma excepcional visto que ndo é possivel declarar fungcbes em G-
Portugol que recebem um ndmero variavel de argumentos.

4.5.2 A funcao “leia”

A funcao “leia” espera por uma entrada do teclado seguida do caractere LF (em geral, associado a tecla
“Enter”) e retorna o valor lido. Esse valor é convertido implicitamente para o tipo do Ivalue. A tabela 4.2
apresenta conversdes implicitas processadas pela fungao “leia” para o enunciado de exemplo “x := leia()”.

Nota: numeros reais sao arredondados, se necessario, e exibidos no formato .xx (com duas casas
decimais). Portanto, o nimero “250.0” ou “250” seria impresso como “250.00” e “1.449” seria impresso
como “1.45”. Se o parametro for uma variavel, seu valor ndo sera modificado.

Conversoes da funcao “leia”
Tipo do LValue “x” | Texto lido Valor final de “x”
inteiro 123 123
inteiro 123 456 123
inteiro abc 0
inteiro 123s 123
I6gico falso falso
l6gico “0” falso
l6gico falso 12wtc | verdadeiro
l6gico 0 umdois3 | verdadeiro

Tabela 4.2: Conversdes implicitas da fungao “leia”

Essa funcdo se comporta de forma excepcional visto que seu tipo de retorno nao é absoluto (over-
loaded, depende do tipo de Ivalue) e, no momento, é restrita expressdes isentas de operadores. Isso
€, ndo é permitido aplica-la como um termo em uma expressdo com multiplos operandos (ex. “X =y +
leia()”), embora seja possivel utiliza-la em subscritos de vetores e matrizes, assim como em expressoes de
estruturas de controle.

Em principio, pode-se levar em conta duas formas de implementar uma fungao de leitura em alto nivel.
A primeira, e mais simples é o uso de fungbes sobrecarregadas, que nao retornam valor e recebem um
parametro de um dado tipo (inteiro, real, etc), que é alimentado com o valor lido. Essa forma exige que a
passagem seja feita por referéncia. A outra forma, é a implementada em G-Portugol, onde a fungao nao
recebe parametros e retorna o valor lido.

As duas formas exigem comportamentos excepcionais, visto que G-Portugol ndo tem suporte para
sobrecarga de funcdes (por parametro ou retorno) ou passagem de parametros por referéncia.

O programa GPT

5.1 Introducao

GPT ¢ a ferramenta principal da linguagem G-Portugol. Entre suas fung¢des principais estao:
e Compilar algoritmos;
e Traduzir algoritmos para outras linguagens;
e Executar algoritmos.

Na versao atual é possivel compilar algoritmos para sistemas Windows e Unices que suportam o for-
mato ELF e o assembler NASM. Também, ha suporte apenas para a traducdo de algoritmos para a lin-
guagem C. O resultado € um codigo em C 100% equivalente ao algoritmo.

E possivel, também, executar algoritmos sem gerar cédigo ou arquivos extra, de forma interpretada.
Também, com a interpretacao, é possivel depurar passo a passo o algoritmo com o uso de uma ferramenta
auxiliar (um cliente de depuragéo).

Vale ressaltar que, qualquer que seja o modo de uso do algoritmo (compilado, interpretado ou traduzido),
se espera que eles tenham sempre comportamentos equivalentes. Isso €, o resultado de uma execugéo
interpretada deve ser o mesmo se a execugao fosse feita por meio de um binario compilado.

5.2 Opcoes gerais
Ao executar o programa “gpt” com o argumento “-h”, € mostrado:

Modo de uso: gpt [opgdes] arquivos

Opgdes:
-v mostra versdo do programa
-h mostra esse texto

-o <arquivo> compila e salva executavel como <arquivo>

-t <arquivo> salva o cdédigo em linguagem C como <arquivo>

-s <arquivo> salva o cddigo em linguagem assembly como <arquivo>
-i interpreta

-d exibe dicas no relatdério de erros

Maiores informacgdes no manual.
As op¢des sdo comentadas a seguir.

e v: Exibe a versdo do programa “gpt”, assim como informagdes de copyright;

h: exibe todas as opc¢des suportadas pelo programa “gpt”;

0 <arquivo>: Ao compilar um algoritmo, salva o executavel com o nome de <arquivo>. Se essa op¢ao
for omitida na compilacdo, o nome do algoritmo sera usado para criar 0 arquivo executavel.

t <arquivo>: Traduz o algoritmo para C, salva o codigo fonte com o nome de <arquivo>;

s <arquivo>: Compila o algoritmo mas nao cria codigo executavel. Salva o cédigo em assembly com
0 nome de <arquivos;

i: Executa o algoritmo diretamente, sem compilar ou criar arquivos. Opgao conhecida como “inter-
pretagdo” ou “scripting”.

e d: Exibe mais informacdes no relatério de erros, como dicas de como proceder para solucionar errors
de sintaxe.

A Ultima opgao (“arquivos”) é uma lista de arquivos contendo o codigo fonte em G-Portugol, embora
seja mais comum utilizar um arquivo apenas para cada algoritmo.

5.3 Tratamento de erros

Em geral, as ferramentas de diversas linguagens de programacdo oferecem o minimo de informacoes
a cerca de erros de compilagdo, as vezes, tendo uma forma “criptica” dificultando seu entendimento e
posterior correcao. Comum, também a possibilidade de que determinados erros sejam reportados em
localizagdes distantes de onde o erro efetivamente se encontra no codigo fonte. Ademais, é importante
notar que nessas linguagens, os erros sao reportados em inglés.

O tratamento de erros é um aspecto importante do programa GPT. A reportagem de erros deve ser o
mais claro possivel e ter um formato uniforme, informando o local exato onde o erro foi localizado. Deve,
também, fornecer dicas de como proceder para corrigir 0 erro, em alguns casos.

No momento, o analisador do GPT percorre o cédigo fonte a procura de erros e, mesmo que encontre,
continua a andlise a procura de mais erros. Uma outra abordagem, seria interromper a analise assim que
um erro fosse encontrado.

5.4 Execucao de programas

5.4.1 Compilacao e geracao de coédigo executavel

O programa GPT é capaz de gerar codigo executavel para arquiteturas compativeis com x86 e em dois
formatos: ELF (Executable and linking format) e PE (Portable Executable). Sistemas Unix, em geral,
suportam o formato ELF, e o formato PE é conhecido nos sistemas Microsoft Windows. Apds o processo
de andlise, o compilador gera cédigo em Assembly para, entdo, usar o NASM (Netwide Assembler) como
backend para montar e criar um executavel valido. Consequentemente, ndo existe etapa de linkagem. A
fase de otimizacao de codigo também néo foi implementada.

Para usar esse recurso, € necessario que o NASM esteja instalado no sistema. Ele pode ser encontrado
em http://www.sf.net/projects/nasm.

5.4.2 Traducao para a linguagem C

E possivel usar o GPT para traduzir algoritmos escritos em G-Portugol para linguagem C. O cédigo resul-
tante, entretanto, ndo é criado com o intuito de ser facilmente lido, visto que a traducao nao é direta. Isso é,
o comportamento esperado pelo algoritmo deve refletir o comportamento que o cédigo em C. Sendo assim,
codigos extras sdo adicionados no arquivo resultante, para, por exemplo, permitir passagem de matrizes
por valor, inicializagdo automatica de variaveis e outras abstragoes.

A tradugao para C é limitada no que diz respeito a nomes e identificadores. E possivel, por exemplo,
declarar uma variavel em G-Portugol com o nome de printf. Ao traduzir para C e tentar compilar o cédigo
resultante, o compilador pode emitir avisos e erros, visto que printf € uma funcao da biblioteca padrao,
usada no cédigo C resultante. Da mesma forma, identificadores com underlines antes ou depois em
seus nomes (como “__leia_texto”) devem ser evitados, pois muitos identificadores internos utilizam essas
convencdes de nomeacgao, e seu uso pode acarretar em conflitos durante a traducao.

5.4.3 Interpretacao de cédigo

O programa GPT permite que o algoritmo seja executado sem gerar cddigo binario. Esse modo é conhecido
como “interpretacao” e linguagens como Perl, PHP e Ruby utilizam esta técnica.

Esse modo permite depurar algoritmos passo a passo (por meio de um client debugger como o GPTE-
ditor) e inspecionar variaveis e a pilha de fungdes enquanto o algoritmo esta em execugao.

A Unica diferenca na execugao de algoritmos em modo interpretado em relagao a outros modos € que
as matrizes/vetores tem seus subscritos checados (“bound checking”). Isso é, erros de execug¢dao sao

emitidos se um indice ndo existir em uma matriz/vetor.

Depuracao interativa

A depuragéo interativa é feita em modo “interpretacdo”, portanto ndo gera codigo binario, executando o
algoritmo diretamente. Para depurar interativamente um algoritmo é necessario um programa extra: o
client debugger. Atualmente, o programa GPTEditor suporta a depuracao interativa.

Entre os recursos disponiveis, pode-se citar a execu¢ao passo a passo em 3 modos (comumente con-
hecidas como “step into”, “step over” e “step out”), inspecao de variaveis locais/globais e pontos de parada
(“breakpoints”).

A depuracgéo ocorre tendo o programa GPT se comunicando via socket com o cliente (ex: GPTEditor),
iniciando transmiss@o de dados entre esses dois pontos. As informacdes enviadas pelo GPT (variaveis,
breakpoints, etc) usam o formato baseado em XML, enquanto o cliente envia comandos simples (0 que
evita que o programa GPT necessite de um XML parser).

Por exemplo, a pilha de funcbes pode ser representada da seguinte forma:

<stackinfo>
<entry id="0" function="@global" line="10"/>
<entry id="1" function="funcTeste" line="18"/>
</stackinfo>

5.4.4 Processando algoritmos divididos em multiplos arquivos

A partir da versao 1.0, o GPT suporta processar algoritmos divididos em multiplos arquivos. Esse recurso
é possivel utilizando duas formas:

e Passando os arquivos como opg¢des na linha de comando;
e Utilizando a variavel de ambiente GPT_INCLUDE.

A primeira forma é explicada na sessao 5.2. A segunda forma pretende facilitar a utilizagao de fungdes
que devem estar disponiveis por padrdo a cada execugao/compilagdo dos algoritmos. Pode-se definir
a variavel de ambiente GPT_INCLUDE contendo varios caminhos de arquivos separados por “”. Em
sistemas Unix, por exemplo, pode-se criar a essa variavel da seguinte forma (utilizando Bash):

$ export GPT_INCLUDE="/usr/local/lib/gpt/base.gpt:/usr/local/lib/gpt/util.gpt"

Onde “base.gpt” e “util.gpt” sdo arquivos contendo fungdes escritas em G-Portugol.

Os arquivos passados pela linha de comando e/ou que se encontram na variavel GPT_INCLUDE sao
concatenados e processados como se 0 algoritmo estivesse em apenas um arquivo. Portanto, arquivos
extras ndo devem ter declaracédo de algoritmo, bloco de variaveis globais ou bloco principal.

Apéndice A

Gramatica da linguagem G-Portugol

A.1 Termos léxicos

A seguir é apresentado as convengdes léxicas usadas em G-Portugol.

Regras para identificar literais numéricos
T_INT_LIT : T_OCTAL_LIT | T_HEX_LIT | T_BIN_LIT | T_DEC_LIT
T_DEC_LIT : [0-9]+
T_OCTAL_LIT : ’0° (°c’|’C’) [0-8]+
T_HEX_LIT : 0’ (Cx’[’X’) [0-9a-fA-F]+
T_BIN_LIT : >0’ (°b’|’B?) [01]+
T_REAL_LIT : T_DEC_LIT+ ’.’ T_DEC_LIT+
Regras para identificar caracteres e cadeias de caracteres
T_CARAC_LIT : 222 ( ~(?22]7\?) [’\? . )7 27
T_STRING_LIT : >"> ( ~C "2 | »\? | CR | LF ) | ?\? . )% "™’
Regras para identificar comentarios
SL_COMMENT : "//" ["LFl* (’\n’)?
ML_COMMENT : "/*" ( “(7%2) | %2 72/7 )x "x/"
Regra para identificar nomes de variaveis, fungdes, etc.

T_IDENTIFICADOR : [a-zA-Z_] [a-zA-Z0-9_]%

A tabela A.1 contém as palavras-chave padréo da linguagem G-Portugol.

A.2 Gramatica

A seguir é apresentado a gramatica da linguagem G-Portugol.

G-Portugol

Palavras-chave de G-Portugol
fim-varidveis algoritmo varigveis inteiro  real
caractere literal l6gico inicio verdadeiro
falso fim ou e nao
se senao entdo fim-se  enquanto
faca fim-enquanto para de até
fim-para matriz inteiros reais caracteres
literais l6gicos funcao retorne passo

Tabela A.1: Palavras-chave de G-Portugol

algoritmo
: declaracao_algoritmo (var_decl_block)? stm_block (func_decls)* EOF

declaracao_algoritmo
"algoritmo" T_IDENTIFICADOR ";"

var_decl_block
"variaveis" (var_decl ";")+ "fim-variaveis"

var_decl
: T_IDENTIFICADOR ("," T_IDENTIFICADOR)* ":" (tp_primitivo | tp_matriz)

tp_primitivo
"inteiro"
llrealll

|

| "caractere"
| "literal"

| "légico"

2

tp_matriz
"matriz" ("[" T_INT_LIT "]")+ "de" tp_prim_pl

tp_prim_pl
: "inteiros"
"reais"
"caracteres"

|
|
| "literais"
| "légicos"

5

stm_block
"inicio" (stm_list)* "fim"

stm_list
stm_attr

| fcall ";"
| stm_ret
| stm_se
| stm_enquanto
| stm_repita
| stm_para

5
stm_ret

"retorne" expr? ";"

lvalue
: T_IDENTIFICADOR ("[" expr "]")x

stm_attr
lvalue ":=" expr ";"

stm_se
"se" expr "ent&do" stm_list ("sen#fo" stm_list)? "fim-se"

stm_enquanto
"enquanto" expr "faga" stm_list "fim-enquanto"

stm_repita
"repita" stm_list "até" expr ";"

stm_para
"para" lvalue '"de" expr "até" expr passo? "faga" stm_list "fim-para"

passo
: "passo" ("+"|"-")7 T_INT_LIT
expr
: expr ("ou"|"[|") expr
expr ("e"|"&&") expr
expr "|" expr

expr """ expr
expr "&" expr

expr (n=u | ||<>u) expr

expr (ll>ll | ny=n | ngn | ||<=u) expr

expr (ll+ll | |l_l|) expr

expr (ll/ll | ll*H | ll%ll) expr

(n+n | n_n | n~n | "néo")? termo
termo

fcall

| 1lvalue
| literal
| " (ll eXpr ll)"

s

fcall
T_IDENTIFICADOR " (" fargs? ")"

fargs
: expr ("," expr)*

literal

: T_STRING_LIT
T_INT_LIT
T_REAL_LIT
T_CARAC_LIT
T_KW_VERDADEIRO
T_KW_FALSO

func_decls
"fung&o" T_IDENTIFICADOR " (" fparams? ")" (":" tb_primitivo)?
fvar_decl
stm_block

fvar_decl
(var_decl ";")*

fparams
: fparam ("," fparam)*
fparam
: T_IDENTIFICADOR ":" (tp_primitivo | tp_matriz)

